

Laboratory 4

Structural & Sequential Systems

Computer Architecture

A.Y. 2023/24


Laboratory goals

- Understand hierarchical and structural approaches to VHDL
- Implement several logic functions in VHDL
- Create a VHDL program to describe flip-flops


Hierarchy & structural analogy

VHDL's structural design is analogous to interconnecting logic devices with wires.

- Top-level entity: can be seen as the whole circuit board.
- VHDL component: represents a reusable logic function
- Signal: wired connection between components.



(a) Hardware implementation with fixed-function logic



(b) VHDL structural implementation

Hierarchical design pattern

Starting from top-level entity, components can have lower-level components inside them.


- When a component is included in a higher-level component, an instance of it is created.
- Allows for design reutilization and nesting.
- Facilitates group work by having each person work on a different subsystem.

Structural synthesis

In the ARCHITECTURE part of an entity, we can create instances of other components.

Syntax:

```
1  ENTITY entity_name IS
2      PORT (input and output definitions);
3  END entity_name;
4
5  ARCHITECTURE arch OF entity_name IS
6      component declarations;
7      signal declarations;
8  BEGIN
9      component instantiations;
10     other statements;
11 END arch;
```



```
COMPONENT component_name IS
    PORT (port definitions);
END COMPONENT component_name
```

Structural synthesis

A VHDL file can use another VHDL file as a component.

- Component declarations resemble entity declarations.
- Each instance of a component requires an instantiation statement.

Structural synthesis - Example

Suppose we have the following AND and OR gates defined as VHDL entities:


```
1  ENTITY and2local IS
2      PORT (i1, i2: IN BIT;
3            o: OUT BIT);
4  END and2local;
5
6  ARCHITECTURE gate OF and2local IS
7  BEGIN
8      o <= i1 AND i2;
9  END gate;
```

```
1  ENTITY or2local IS
2      PORT (i1, i2: IN BIT;
3            o: OUT BIT);
4  END or2local;
5
6  ARCHITECTURE gate OF or2local IS
7  BEGIN
8      o <= i1 OR i2;
9  END gate;
```

Note: these are for simplicity's sake; you don't need to create entities for primitive logic functions in VHDL!

Structural synthesis – Example 1

We want to reuse these components to create the following circuit:



Structural synthesis – Example 1

```
1  USE work.all;
2
3  entity main_system is
4      port (in1, in2, in3, in4: IN BIT;
5            out3: OUT BIT);
6  end main_system;
7
8  architecture internal of main_system is
9      component and2local is
10         port (i1, i2: IN BIT;
11              o: OUT BIT);
12     end component;
13
14     component or2local is
15         port (i1, i2: IN BIT;
16              o: OUT BIT);
17     end component;
18
19     signal out1, out2: BIT;
20
21 begin
22     G1: and2local
23         port map(i1 => in1,
24                 i2 => in2,
25                 o => out1);
26     G2: and2local
27         port map(i1 => in3,
28                 i2 => in4,
29                 o => out2);
30     G3: or2local
31         port map(i1 => out1,
32                 i2 => out2,
33                 o => out3);
34 end internal;
```

USE work.all; → Use the work library, which allows for component reutilization in VHDL projects.

Component declarations

Component instantiations

Structural synthesis – Example 2

```
2 ENTITY FUNCTION1 IS
3   PORT( X: IN BIT_VECTOR (2 downto 0);
4         z: OUT BIT);
5 END FUNCTION1;
6
7 ARCHITECTURE AND_OR of FUNCTION1 IS
8
9   COMPONENT AND2LOCAL
10    PORT (I1, I2: IN BIT;
11          O: OUT BIT);
12 END COMPONENT;
13
14   COMPONENT OR3LOCAL
15    PORT (I1, I2, I3: IN BIT;
16          O: OUT BIT);
17 END COMPONENT;
18
19   SIGNAL A1, A2, A3: BIT;
20
21 BEGIN
22
23   l1:    AND2LOCAL
24         port map(X(0), X(1), A1);
25
26   l2:    AND2LOCAL
27         port map(X(0), X(2), A2);
28
29   l3:    AND2LOCAL
30         port map(X(1), X(2), A3);
31
32   l4:    OR3LOCAL
33         port map(A1, A2, A3, Z);
34 END AND_OR;
```


Component declarations

(they refer to these separate files!)

```
ENTITY AND2LOCAL IS
  PORT(I1, I2: IN BIT;
        O: OUT BIT);
END AND2LOCAL;
ARCHITECTURE gate OF AND2LOCAL IS
BEGIN
  O <= I1 and I2;
END gate;
```


```
ENTITY OR3LOCAL IS
  PORT(I1, I2, I3: IN BIT;
        O: OUT BIT);
END OR3LOCAL;
ARCHITECTURE gate OF OR3LOCAL IS
BEGIN
  O <= I1 or I2 or I3;
END gate;
```

Component instantiations



Exercise 1


- Create a new project and add the VHDL files from lab4_struc_examples.zip
- Based on Example 2, implement the component “*function2*” following this schematic;
- Next, use components *function1* and *function2* component to compile the VHDL file “the_bigger.vhd”



Exercise 1

the_bigger.vhd:

```
1  ENTITY the_bigger IS
2      PORT( A: IN BIT_VECTOR (2 downto 0);
3            C: OUT BIT_VECTOR (2 downto 0));
4  end the_bigger;
5
6  ARCHITECTURE structural of the_bigger IS
7
8      COMPONENT function1 --local
9          PORT( X: IN BIT_VECTOR (2 downto 0);
10              Z: OUT BIT);
11      END COMPONENT;
12
13      COMPONENT function2 --local
14          PORT( X: IN BIT_VECTOR (2 downto 0);
15              Z: OUT BIT);
16      END COMPONENT;
17
18  BEGIN
19
20      COM1:  function1
21          port map(A, C(1));
22
23      COM2: FUNCTION2
24          port map (A, C(0));
25  END structural;
```




Part 2

Sequential systems in VHDL

Sequential circuits in VHDL

- LATCH (\overline{SET} – \overline{RESET})

```
1  LIBRARY ieee;
2  USE ieee.std_logic_1164.all;
3
4  ENTITY srlatch is
5      PORT(      set_not, reset_not: IN STD_LOGIC;
6                q, q_not: INOUT STD_LOGIC);
7
8      END srlatch;
9
10 ARCHITECTURE logic_operation OF srlatch IS
11 BEGIN
12     q      <= q_not nand set_not;
13     q_not  <= q nand reset_not;
14 END logic_operation;
```



Truth table for an active-LOW input \overline{S} - \overline{R} latch.

Inputs		Outputs		Comments
\overline{S}	\overline{R}	Q	\overline{Q}	
1	1	NC	NC	No change. Latch remains in present state.
0	1	1	0	Latch SET.
1	0	0	1	Latch RESET.
0	0	1	1	Invalid condition

VHDL Processes

A process consists of a set of instructions analyzed sequentially. It behaves as a single instruction, and it is used to perform algorithmic or behavioral descriptions.

It has a declaration section for variables, types, constants and subprograms. The sequential instructions include *if*, *case*, *loop*, *next*, *assert*, *wait* statements.

Sensitivity lists: trigger the process.

Syntax:


```
process_label:
PROCESS(sensitivity list)
    variable declaration;
BEGIN
    sequential instructions;
END PROCESS process_label;
```

VHDL Processes

- Sensitivity lists:

They are made up of defined signals that trigger process execution upon change.

Process will trigger
when a or b change.



```
3  LIBRARY ieee;
4  USE ieee.std_logic_1164.ALL;
5  ▾ ENTITY compare8 IS
6  ▾      PORT(
7          a,b : IN STD_LOGIC_VECTOR (7 downto 0);
8          agtb, aeqb, altb: OUT STD_LOGIC);
9  END compare8;
10 ▾ ARCHITECTURE a OF compare8 IS
11     SIGNAL compare: STD_LOGIC_VECTOR (2 downto 0);
12 ▾ BEGIN
13     → PROCESS (a,b)
14     BEGIN
15         IF a<b THEN
16             compare <= "110";
17         ELSIF a=b THEN
18             compare <= "101";
19         ELSIF a>b THEN
20             compare <= "011";
21         ELSE
22             compare <= "111";
23         END IF;
24         agtb <= compare(2);
25         aeqb <= compare(1);
26         altb <= compare(0);
27     END PROCESS;
28 END a;
```



Sequential circuits in VHDL

- LATCH-D

```
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.ALL;
4  ENTITY latchff IS
5      PORT(
6          d,clk : IN STD_LOGIC;
7          q: OUT STD_LOGIC);
8  END latchff;
9  ARCHITECTURE behavior OF latchff IS
10 BEGIN
11     PROCESS (d,clk)
12     BEGIN
13         IF clk = '1' THEN
14             q <= d;
15         END IF;
16     END PROCESS;
17 END behavior;
```

Sequential circuits in VHDL

- D-Flipflop (rising edge).



Tells the compiler to check
on the rising edge of the
clock signal.

```
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4
5  ENTITY d_ff IS
6      PORT(
7          D, Clock: IN STD_LOGIC;
8          Q       : OUT STD_LOGIC);
9  END d_ff;
10
11 ARCHITECTURE behavior OF d_ff IS
12 BEGIN
13     PROCESS(Clock)
14     BEGIN
15         IF Clock'EVENT and Clock='1' THEN
16             Q <= D;
17         END IF;
18     END PROCESS;
19 END behavior;
```

Sequential circuits in VHDL

- D-Flipflop with reset (rising edge).

Synchronous

```
LIBRARY ieee;
USE ieee.std_logic_1164.all;

ENTITY d_ffs IS
    PORT(
        D, Reset, Clock: IN STD_LOGIC;
        Q: OUT STD_LOGIC);
END d_ffs;

ARCHITECTURE behavior OF d_ffs IS
BEGIN
    PROCESS
    BEGIN
        WAIT UNTIL Clock'EVENT and Clock='1';
        IF Reset = '0' THEN
            Q <= '0';
        ELSE
            Q <= D;
        END IF;
    END PROCESS;
END behavior;
```

Asynchronous

```
LIBRARY ieee;
USE ieee.std_logic_1164.all;

ENTITY d_ffa IS
    PORT(
        D, Reset, Clock: IN STD_LOGIC;
        Q: OUT STD_LOGIC);
END d_ffa;

ARCHITECTURE behavior OF d_ffa IS
BEGIN
    PROCESS(Reset,Clock)
    BEGIN
        IF Reset='0' THEN
            Q <= '0';
        ELSIF Clock'EVENT and Clock='1' THEN
            Q <= D;
        END IF;
    END PROCESS;
END behavior;
```

VHDL CASE statement


The CASE statement is used to execute one of several instruction, based on a signal's value.

```
CASE __expression IS
  WHEN __constant_value =>
    __statement;
    __statement;
  WHEN __constant_value =>
    __statement;
    __statement;
  WHEN OTHERS =>
    __statement;
    __statement;
END CASE;
```

```
CASE s IS
  WHEN "00" =>
    y <= "0001";
    x <= "1110";
  WHEN "01" =>
    y => "0010";
    x => "1101";
  WHEN "10" =>
    y <= "0100";
    x <= "1011";
  WHEN "11" =>
    y <= "1000";
    x <= "0111";
  WHEN others =>
    y <= "0000";
    x <= "1111";
END CASE;
```

Sequential circuits in VHDL

- J-K Flipflop (rising edge)




```
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;
ENTITY jk_ff IS
    PORT(
        j,k,clk : IN STD_LOGIC;
        q: BUFFER STD_LOGIC);
END jk_ff;
ARCHITECTURE behavior OF jk_ff IS
BEGIN
    PROCESS
        variable jk:std_logic_vector (1 downto 0);
    BEGIN
        WAIT UNTIL clk='1';
        jk := j&k;
        CASE (jk) IS
            WHEN "01" =>
                q <= '0';
            WHEN "10" =>
                q <= '1';
            WHEN "11" =>
                q <= not q;
            WHEN others =>
                NULL;
        END CASE;
    END PROCESS;
END behavior;
```

Exercise 2

- Create a new project and add the VHDL files from lab4_seq_examples.zip
- Create a **4-bit Up-Counter**. Use flip-flops as components, and only use PROCESS constructs.
- You are free to use any flip-flop from the example files.

Exercise 2 – Sample using D-flipflops



Source: https://home.engineering.iastate.edu/alexs/classes/2018_Fall_281/slides_PDF/31_Solved_Problems.pdf

References

- Tokheim, R., *Digital Electronics: Principles And Applications*.
- Floyd, Thomas L., *Digital Fundamentals*.
- Mano, Morris. *Digital Design*.
- Intel QUARTUS Help Manuals
- Tocci, Ronald. *Digital Systems Principles and Applications*.